

(* Asymptotic expected number of hairpins in saturated structures with theta=1 and p=1 *)

(*We first compute the dominant singularity and asymptotic number of saturated structures, using Drmota–Lalley–Woods Theorem. *)

```

Clear["*"]
Clear[p, S, D0, N0, z, R, eqn0, eqn, F, z0, y0, dFdzOfz0S0, d2FdyOfz0S0];
p = 1;
eqn = {S == D0 + N0, D0 == z + z^2, N0 == R D0 + p D0 z^2 + p N0 z^2 + p S D0 z^2 + p S N0 z^2,
  R == p D0 z^2 + p N0 z^2 + p R D0 z^2 + p R N0 z^2}
Eliminate[eqn, {N0, D0, R}]
F = (S^3 z^4 + S^2 z^2 (-2 + z^2) - z (1 + z)) / (z^2 - 1)

NSolve[{F == S, D[F, S] == 1}, {z, S}];
z0 = 0.42468731042025953`
y0 = 1.6568963458689725`

dFdzOfz0S0 = D[F, z] /. {z -> z0, S -> y0}

d2FdyOfz0S0 = D[F, {S, 2}] /. {z -> z0, S -> y0}
c = Sqrt[z0 dFdzOfz0S0 / (2 Pi d2FdyOfz0S0)]
c * (1 / z0) ^ n n ^ (-3 / 2)

{S == D0 + N0, D0 == z + z^2, N0 == D0 R + D0 z^2 + N0 z^2 + D0 S z^2 + N0 S z^2, R == D0 z^2 + N0 z^2 + D0 R z^2 + N0 R z^2}
S^3 z^4 + S (1 - z^2) + S^2 z^2 (-2 + z^2) == z (1 + z)
S^3 z^4 - z (1 + z) + S^2 z^2 (-2 + z^2)
-----
-1 + z^2
0.424687
1.6569
6.93659
0.406264
1.07427
1.07427 x 2.35467^n
-----
n^3/2

```

(*Now, we compute mean, variance using Drmota's Theorem*)

```

Clear["*"]
p = 1;
eqn = {S == D0 + N0, D0 == z + z^2, N0 == R D0 + p u D0 z^2 + p N0 z^2 + p u S D0 z^2 + p S N0 z^2,
  R == p u D0 z^2 + p N0 z^2 + p u R D0 z^2 + p R N0 z^2}

```

```

CellPrint["Eliminate all variables except S,u,z"]
Eliminate[eqn, {D0, N0, R}]
Collect[%, S, Simplify]
F = (S^3 z^4 + S^2 z^2 (-2 + z^2 + 2 (-1 + u) z^3 + 2 (-1 + u) z^4) -
      (-z (1 + z) (-1 - (-1 + u) z^2 + (-1 + u)^2 z^5 + (-1 + u)^2 z^6))) /
      (- (1 + z) (1 - z - 2 (-1 + u) z^3 + 2 (-1 + u) z^5 + (-1 + u)^2 z^6 + (-1 + u)^2 z^7))
f = (F /. S -> s)
s - f
(* express over a common denominator*)
Together[s - f]
a = Numerator[%]

dfs = D[f, s]
1 - dfs
(* express over a common denominator*)
Together[1 - dfs]
(* a is numerator of s-f, and b is numerator of 1-dfs,
where both have identical denominators *)
b = Numerator[%]
(* a is numerator of s-f, and b is numerator of 1-dfs,
where both have identical denominators *)

(* NOTE: denominators of both expressions s-f and 1-dfs is the same *)
CellPrint["Now compute the resultant of numerators a,b to get relation between S,u,z"]

(* res =Resultant[s-f,1-dfs,s] *)

(*We compute resultant of numerators,
since we have Resultant[S-F,1-D[F,S],S]=0 which is not much use.*)
CellPrint["Express S-F and 1-D[F,S] as fractions over the same common denominator"]
CellPrint["Then compute the resultant of the numerators of these expressions"]
res = Resultant[a, b, s]
(* Replace z by z[u], a function of u *)

res /. z -> z[u]
(* Now compute z'[u] *)
dres = D[%, u]
Simplify[Collect[dres, z'[u]]]
Solve[dres == 0, z'[u]]
dzu = Last[Last[Last[Solve[dres == 0, z'[u]]]]];
(*z[1] equals rho, the dominant singularity *)

rho = 0.42468731042025953` ;
(* value of z0 in the first part of this file, the dominant singularity*)
dzuEvaluatedAt1 = (dzu /. u -> 1) /. z[1] -> rho
CellPrint[
  "According to Drmota's Theorem 1, the mean equals -z'[1]/z[1], computed next. "]

mu = ((-dzu / z[u] /. u -> 1) /. z[1] -> rho)

(*Variance computation *)
(* d2zu is z''[1] *)
d2zu = ((D[dzu, u] /. u -> 1) /. z[1] -> rho) /. z'[1] -> dzuEvaluatedAt1
CellPrint["Now compute variance, which by Drmota is -z''[1]/z[1] + mu^2 + mu"]
var = -d2zu / rho + mu * mu + mu

```

$$\{S == D0 + N0, D0 == z + z^2, \\ N0 == D0 R + N0 z^2 + N0 S z^2 + D0 u z^2 + D0 S u z^2, R == N0 z^2 + N0 R z^2 + D0 u z^2 + D0 R u z^2\}$$

Eliminate all variables except S,u,z

$$\begin{aligned} & S^3 z^4 + S^2 z^2 (-2 + z^2 - 2 z^3 + 2 u z^3 - 2 z^4 + 2 u z^4) + S (1 - z^2 + 2 z^3 - 2 u z^3 + \\ & \quad 2 z^4 - 2 u z^4 - 2 z^5 + 2 u z^5 - z^6 + u^2 z^6 + 2 z^7 - 4 u z^7 + 2 u^2 z^7 + z^8 - 2 u z^8 + u^2 z^8) == \\ & z (1 + z - z^2 + u z^2 - z^3 + u z^3 - z^5 + 2 u z^5 - u^2 z^5 - 2 z^6 + 4 u z^6 - 2 u^2 z^6 - z^7 + 2 u z^7 - u^2 z^7) \\ & S^3 z^4 + S^2 z^2 (-2 + z^2 + 2 (-1 + u) z^3 + 2 (-1 + u) z^4) + \\ & \quad S (1 + z) (1 - z - 2 (-1 + u) z^3 + 2 (-1 + u) z^5 + (-1 + u)^2 z^6 + (-1 + u)^2 z^7) == \\ & -z (1 + z) (-1 - (-1 + u) z^2 + (-1 + u)^2 z^5 + (-1 + u)^2 z^6) \\ & (S^3 z^4 + S^2 z^2 (-2 + z^2 + 2 (-1 + u) z^3 + 2 (-1 + u) z^4) + \\ & \quad z (1 + z) (-1 - (-1 + u) z^2 + (-1 + u)^2 z^5 + (-1 + u)^2 z^6)) / \\ & ((-1 - z) (1 - z - 2 (-1 + u) z^3 + 2 (-1 + u) z^5 + (-1 + u)^2 z^6 + (-1 + u)^2 z^7)) \\ & (s^3 z^4 + s^2 z^2 (-2 + z^2 + 2 (-1 + u) z^3 + 2 (-1 + u) z^4) + \\ & \quad z (1 + z) (-1 - (-1 + u) z^2 + (-1 + u)^2 z^5 + (-1 + u)^2 z^6)) / \\ & ((-1 - z) (1 - z - 2 (-1 + u) z^3 + 2 (-1 + u) z^5 + (-1 + u)^2 z^6 + (-1 + u)^2 z^7)) \\ & s - (s^3 z^4 + s^2 z^2 (-2 + z^2 + 2 (-1 + u) z^3 + 2 (-1 + u) z^4) + \\ & \quad z (1 + z) (-1 - (-1 + u) z^2 + (-1 + u)^2 z^5 + (-1 + u)^2 z^6)) / \\ & ((-1 - z) (1 - z - 2 (-1 + u) z^3 + 2 (-1 + u) z^5 + (-1 + u)^2 z^6 + (-1 + u)^2 z^7)) \\ & (s - z - z^2 - s z^2 - 2 s^2 z^2 + z^3 + 2 s z^3 - u z^3 - 2 s u z^3 + z^4 + 2 s z^4 + s^2 z^4 + s^3 z^4 - u z^4 - 2 s u z^4 - 2 s z^5 - \\ & \quad 2 s^2 z^5 + 2 s u z^5 + 2 s^2 u z^5 + z^6 - s z^6 - 2 s^2 z^6 - 2 u z^6 + 2 s^2 u z^6 + u^2 z^6 + s u^2 z^6 + 2 z^7 + \\ & \quad 2 s z^7 - 4 u z^7 - 4 s u z^7 + 2 u^2 z^7 + 2 s u^2 z^7 + z^8 + s z^8 - 2 u z^8 - 2 s u z^8 + u^2 z^8 + s u^2 z^8) / \\ & ((1 + z) (1 - z + 2 z^3 - 2 u z^3 - 2 z^5 + 2 u z^5 + z^6 - 2 u z^6 + u^2 z^6 + z^7 - 2 u z^7 + u^2 z^7)) \\ & s - z - z^2 - s z^2 - 2 s^2 z^2 + z^3 + 2 s z^3 - u z^3 - 2 s u z^3 + z^4 + 2 s z^4 + s^2 z^4 + s^3 z^4 - u z^4 - 2 s u z^4 - \\ & \quad 2 s z^5 - 2 s^2 z^5 + 2 s u z^5 + 2 s^2 u z^5 + z^6 - s z^6 - 2 s^2 z^6 - 2 u z^6 + 2 s^2 u z^6 + u^2 z^6 + s u^2 z^6 + \\ & \quad 2 z^7 + 2 s z^7 - 4 u z^7 - 4 s u z^7 + 2 u^2 z^7 + 2 s u^2 z^7 + z^8 + s z^8 - 2 u z^8 - 2 s u z^8 + u^2 z^8 + s u^2 z^8 \\ & \quad 3 s^2 z^4 + 2 s z^2 (-2 + z^2 + 2 (-1 + u) z^3 + 2 (-1 + u) z^4) \\ & \quad \hline & (-1 - z) (1 - z - 2 (-1 + u) z^3 + 2 (-1 + u) z^5 + (-1 + u)^2 z^6 + (-1 + u)^2 z^7) \\ & 1 - (3 s^2 z^4 + 2 s z^2 (-2 + z^2 + 2 (-1 + u) z^3 + 2 (-1 + u) z^4)) / \\ & ((-1 - z) (1 - z - 2 (-1 + u) z^3 + 2 (-1 + u) z^5 + (-1 + u)^2 z^6 + (-1 + u)^2 z^7)) \\ & (1 - z^2 - 4 s z^2 + 2 z^3 - 2 u z^3 + 2 z^4 + 2 s z^4 + 3 s^2 z^4 - 2 u z^4 - 2 z^5 - 4 s z^5 + 2 u z^5 + \\ & \quad 4 s u z^5 - z^6 - 4 s z^6 + 4 s u z^6 + u^2 z^6 + 2 z^7 - 4 u z^7 + 2 u^2 z^7 + z^8 - 2 u z^8 + u^2 z^8) / \\ & ((1 + z) (1 - z + 2 z^3 - 2 u z^3 - 2 z^5 + 2 u z^5 + z^6 - 2 u z^6 + u^2 z^6 + z^7 - 2 u z^7 + u^2 z^7)) \\ & 1 - z^2 - 4 s z^2 + 2 z^3 - 2 u z^3 + 2 z^4 + 2 s z^4 + 3 s^2 z^4 - 2 u z^4 - 2 z^5 - 4 s z^5 + \\ & \quad 2 u z^5 + 4 s u z^5 - z^6 - 4 s z^6 + 4 s u z^6 + u^2 z^6 + 2 z^7 - 4 u z^7 + 2 u^2 z^7 + z^8 - 2 u z^8 + u^2 z^8 \end{aligned}$$

Now compute the resultant of numerators a,b to get relation between S,u,z

Express S-F and 1-D[F,S] as fractions over the same common denominator

Then compute the resultant of the numerators of these expressions

$$\begin{aligned} & -4 z^{11} - 5 z^{12} + 6 z^{13} + 23 z^{14} + 12 u z^{14} + 34 z^{15} + 26 u z^{15} + 12 z^{16} + 20 u z^{16} - 30 z^{17} + 38 u z^{17} - \\ & \quad 12 u^2 z^{17} - 61 z^{18} + 94 u z^{18} - 37 u^2 z^{18} - 74 z^{19} + 124 u z^{19} - 50 u^2 z^{19} - 65 z^{20} + 122 u z^{20} - \\ & \quad 61 u^2 z^{20} + 4 u^3 z^{20} - 52 z^{21} + 120 u z^{21} - 84 u^2 z^{21} + 16 u^3 z^{21} - 36 z^{22} + 96 u z^{22} - 84 u^2 z^{22} + \\ & \quad 24 u^3 z^{22} - 16 z^{23} + 48 u z^{23} - 48 u^2 z^{23} + 16 u^3 z^{23} - 4 z^{24} + 12 u z^{24} - 12 u^2 z^{24} + 4 u^3 z^{24} \end{aligned}$$

$$\begin{aligned}
& -4 z[u]^{11} - 5 z[u]^{12} + 6 z[u]^{13} + 23 z[u]^{14} + 12 u z[u]^{14} + 34 z[u]^{15} + 26 u z[u]^{15} + 12 z[u]^{16} + \\
& 20 u z[u]^{16} - 30 z[u]^{17} + 38 u z[u]^{17} - 12 u^2 z[u]^{17} - 61 z[u]^{18} + 94 u z[u]^{18} - 37 u^2 z[u]^{18} - 74 z[u]^{19} + \\
& 124 u z[u]^{19} - 50 u^2 z[u]^{19} - 65 z[u]^{20} + 122 u z[u]^{20} - 61 u^2 z[u]^{20} + 4 u^3 z[u]^{20} - 52 z[u]^{21} + \\
& 120 u z[u]^{21} - 84 u^2 z[u]^{21} + 16 u^3 z[u]^{21} - 36 z[u]^{22} + 96 u z[u]^{22} - 84 u^2 z[u]^{22} + 24 u^3 z[u]^{22} - \\
& 16 z[u]^{23} + 48 u z[u]^{23} - 48 u^2 z[u]^{23} + 16 u^3 z[u]^{23} - 4 z[u]^{24} + 12 u z[u]^{24} - 12 u^2 z[u]^{24} + 4 u^3 z[u]^{24} \\
& 12 z[u]^{14} + 26 z[u]^{15} + 20 z[u]^{16} + 38 z[u]^{17} - 24 u z[u]^{17} + 94 z[u]^{18} - 74 u z[u]^{18} + \\
& 124 z[u]^{19} - 100 u z[u]^{19} + 122 z[u]^{20} - 122 u z[u]^{20} + 12 u^2 z[u]^{20} + 120 z[u]^{21} - \\
& 168 u z[u]^{21} + 48 u^2 z[u]^{21} + 96 z[u]^{22} - 168 u z[u]^{22} + 72 u^2 z[u]^{22} + 48 z[u]^{23} - 96 u z[u]^{23} + \\
& 48 u^2 z[u]^{23} + 12 z[u]^{24} - 24 u z[u]^{24} + 12 u^2 z[u]^{24} - 44 z[u]^{10} \text{Derivative}[1][z][u] - \\
& 60 z[u]^{11} \text{Derivative}[1][z][u] + 78 z[u]^{12} \text{Derivative}[1][z][u] + 322 z[u]^{13} \text{Derivative}[1][z][u] + \\
& 168 u z[u]^{13} \text{Derivative}[1][z][u] + 510 z[u]^{14} \text{Derivative}[1][z][u] + \\
& 390 u z[u]^{14} \text{Derivative}[1][z][u] + 192 z[u]^{15} \text{Derivative}[1][z][u] + \\
& 320 u z[u]^{15} \text{Derivative}[1][z][u] - 510 z[u]^{16} \text{Derivative}[1][z][u] + \\
& 646 u z[u]^{16} \text{Derivative}[1][z][u] - 204 u^2 z[u]^{16} \text{Derivative}[1][z][u] - \\
& 1098 z[u]^{17} \text{Derivative}[1][z][u] + 1692 u z[u]^{17} \text{Derivative}[1][z][u] - \\
& 666 u^2 z[u]^{17} \text{Derivative}[1][z][u] - 1406 z[u]^{18} \text{Derivative}[1][z][u] + \\
& 2356 u z[u]^{18} \text{Derivative}[1][z][u] - 950 u^2 z[u]^{18} \text{Derivative}[1][z][u] - \\
& 1300 z[u]^{19} \text{Derivative}[1][z][u] + 2440 u z[u]^{19} \text{Derivative}[1][z][u] - \\
& 1220 u^2 z[u]^{19} \text{Derivative}[1][z][u] + 80 u^3 z[u]^{19} \text{Derivative}[1][z][u] - \\
& 1092 z[u]^{20} \text{Derivative}[1][z][u] + 2520 u z[u]^{20} \text{Derivative}[1][z][u] - \\
& 1764 u^2 z[u]^{20} \text{Derivative}[1][z][u] + 336 u^3 z[u]^{20} \text{Derivative}[1][z][u] - \\
& 792 z[u]^{21} \text{Derivative}[1][z][u] + 2112 u z[u]^{21} \text{Derivative}[1][z][u] - \\
& 1848 u^2 z[u]^{21} \text{Derivative}[1][z][u] + 528 u^3 z[u]^{21} \text{Derivative}[1][z][u] - \\
& 368 z[u]^{22} \text{Derivative}[1][z][u] + 1104 u z[u]^{22} \text{Derivative}[1][z][u] - \\
& 1104 u^2 z[u]^{22} \text{Derivative}[1][z][u] + 368 u^3 z[u]^{22} \text{Derivative}[1][z][u] - \\
& 96 z[u]^{23} \text{Derivative}[1][z][u] + 288 u z[u]^{23} \text{Derivative}[1][z][u] - \\
& 288 u^2 z[u]^{23} \text{Derivative}[1][z][u] + 96 u^3 z[u]^{23} \text{Derivative}[1][z][u] \\
& 2 z[u]^{10} (6 (-1 + u)^2 z[u]^{14} - 22 \text{Derivative}[1][z][u] - 30 z[u] \text{Derivative}[1][z][u] + \\
& 39 z[u]^2 \text{Derivative}[1][z][u] + 7 (23 + 12 u) z[u]^3 \text{Derivative}[1][z][u] + \\
& 24 (-1 + u)^2 z[u]^{13} (1 + 2 (-1 + u) \text{Derivative}[1][z][u]) + \\
& 4 (-1 + u) z[u]^{12} (-12 + 9 u + 46 (-1 + u)^2 \text{Derivative}[1][z][u]) + \\
& z[u]^{10} (61 - 61 u + 6 u^2 + 42 (-1 + u)^2 (-13 + 4 u) \text{Derivative}[1][z][u]) + \\
& z[u]^5 (13 + 32 (3 + 5 u) \text{Derivative}[1][z][u]) + 3 z[u]^4 (2 + (85 + 65 u) \text{Derivative}[1][z][u]) + \\
& 12 (-1 + u) z[u]^{11} (-5 + 2 u + 11 (3 - 5 u + 2 u^2) \text{Derivative}[1][z][u]) + \\
& z[u]^6 (10 - 17 (15 - 19 u + 6 u^2) \text{Derivative}[1][z][u]) + \\
& z[u]^8 (47 - 37 u - 19 (37 - 62 u + 25 u^2) \text{Derivative}[1][z][u]) + \\
& z[u]^7 (19 - 12 u - 9 (61 - 94 u + 37 u^2) \text{Derivative}[1][z][u]) + \\
& 2 z[u]^9 (31 - 25 u + 5 (-65 + 122 u - 61 u^2 + 4 u^3) \text{Derivative}[1][z][u])) \\
& \{ \{ \text{Derivative}[1][z][u] \rightarrow \\
& (-6 z[u]^4 - 13 z[u]^5 - 10 z[u]^6 - 19 z[u]^7 + 12 u z[u]^7 - 47 z[u]^8 + 37 u z[u]^8 - \\
& 62 z[u]^9 + 50 u z[u]^9 - 61 z[u]^{10} + 61 u z[u]^{10} - 6 u^2 z[u]^{10} - 60 z[u]^{11} + \\
& 84 u z[u]^{11} - 24 u^2 z[u]^{11} - 48 z[u]^{12} + 84 u z[u]^{12} - 36 u^2 z[u]^{12} - \\
& 24 z[u]^{13} + 48 u z[u]^{13} - 24 u^2 z[u]^{13} - 6 z[u]^{14} + 12 u z[u]^{14} - 6 u^2 z[u]^{14}) / \\
& (-22 - 30 z[u] + 39 z[u]^2 + 161 z[u]^3 + 84 u z[u]^3 + 255 z[u]^4 + 195 u z[u]^4 + 96 z[u]^5 + \\
& 160 u z[u]^5 - 255 z[u]^6 + 323 u z[u]^6 - 102 u^2 z[u]^6 - 549 z[u]^7 + 846 u z[u]^7 - \\
& 333 u^2 z[u]^7 - 703 z[u]^8 + 1178 u z[u]^8 - 475 u^2 z[u]^8 - 650 z[u]^9 + 1220 u z[u]^9 - \\
& 610 u^2 z[u]^9 + 40 u^3 z[u]^9 - 546 z[u]^{10} + 1260 u z[u]^{10} - 882 u^2 z[u]^{10} + 168 u^3 z[u]^{10} - \\
& 396 z[u]^{11} + 1056 u z[u]^{11} - 924 u^2 z[u]^{11} + 264 u^3 z[u]^{11} - 184 z[u]^{12} + 552 u z[u]^{12} - \\
& 552 u^2 z[u]^{12} + 184 u^3 z[u]^{12} - 48 z[u]^{13} + 144 u z[u]^{13} - 144 u^2 z[u]^{13} + 48 u^3 z[u]^{13}) \} \} \\
& -0.0523188
\end{aligned}$$

According to Drmota's Theorem 1, the mean equals $-z'[1]/z[1]$, computed next.

0.123194

0.0442454

Now compute variance, which by Drmota is $-z''[1]/z[1] + \mu^2 + \mu$

0.0341867